Zusatzmaterial zum Buch

Modellierung, Analyse und Simulation elektrischer und mechanischer Systeme mit Maple und MapleSim

Springer Vieweg 2015
Autor: Prof. Dr.-Ing. Rolf Müller

Kreisdiagramm einer Asynchronmaschine mit Maple konstruieren (Variante 1)

Autor: R. Müller
Stand: 15.08.2015

Ein Kreisdiagramm, d. h. die Ortskurve des Ständerstromes, ermöglicht es, das Verhalten eines Drehstrom-Asynchronmotors bei veränderlicher Belastung schnell zu überblicken. Für einen Asynchronmotor mit Schleifringläufer wird es mit Hilfe von Maple konstruiert. Verwendet wird die vereinfachte Form (Heyland-Diagramm). Bezüglich der theoretischen Grundlagen der Konstruktion des Kreisdiagramms wird auf [2] und [3] verwiesen.
Die folgenden Daten [1] werden der Konstruktion zugrunde gelegt, können aber im Programm leicht durch andere ersetzt werden:

$U_n = 500$ V, $f = 50$ Hz, $I_{1,n} = 28{,}7$ A, $N = 18$ kW, $n = 967{,}5$ min^{-1}, $\cos\varphi = 0{,}844$
Rotor: $U_{2,n} = 83{,}2$ V, Strangwiderstände der Wicklungen: $R_1 = 0{,}465\ \Omega$, $R_2 = 0{,}00985\ \Omega$, Stern-/Sternschaltung
Leerlauf: $I_{1,0} = 8{,}5$ A, $\cos\phi_0 = 0{,}15$;
Kurzschlussversuch: $U_k = 170{,}5$ V bei $I_{1,n} = 28{,}7$ A, $\cos\varphi_k = 0{,}277$

Programmbeschreibung
Die Parameter des Kreises werden aus drei Kreispunkten P_0, P_n und P_k, die durch die komplexen Statorströme (Betrag und Winkel) für die Zustände Leerlauf, Nennbetrieb und Kurzschluss definiert werden, berechnet. Diese Punkte der gesuchten Ortskurve liefern unter Verwendung der allgemeinen Kreisgleichung ein System von drei quadratischen Gleichungen zur Bestimmung der Koordinaten des Mittelpunktes und des Radius des Kreises. Der so ermittelte Heylandkreis wird im Maple-Worksheet dargestellt, dann durch die Eintragung des Punktes P_{kipp} ergänzt und ausgewertet.
Die Einführung der Variablen P_0, P_n und P_k zur Bezeichnung der charakteristischen Punkte dient der Vereinfachung der Schreibweise der Gleichungen.

Ein zweiter Lösungsweg zur Konstruktion eines Kreisdiagramms, für den lediglich der Leerlaufpunkt und der Kurzschlusspunkt bekannt sein müssen, wird im Maple-Worksheet "Kreisdiagramm Asynchronmaschine, Bsp. 2.mw" verwendet.

Das Kreisdiagramm mit den Maßstäben für Strom, Leistung und Drehmoment kann schließlich auch ausgedruckt werden. Die dafür nötigen Schritte werden unter 6. beschrieben.

Literatur

[1] Bläss, H.: Definitionsbeispiel zur Kreisdiagrammkonstruktion
[2] Bödefeld/Sequenz: Elektrische Maschinen.7. Aufl., Springer Verlag 1965, S. 212 ff.
[3] Nürnberg, W.: Die Asynchronmaschine. 2. Aufl., Springer-Verlag 1963
[4] Müller, R.: Modellierung, Analyse und Simulation elektrischer und mechanischer Systeme mit Maple und MapleSim.
Springer Vieweg 2015

```
> restart: with(plots):
> interface(imaginaryunit=j, displayprecision=3):
> plotsetup("inline",plotoutput=terminal,plotoptions="colour=cmyk,
  resolution=2000"):
```

Gegebene Motordaten:

Nennbetrieb:

```
> Typ:= "Asynchronmotor mit Schleifringläufer, 18 kW":
> U[n]:= 500:    # V
> U[2,n]:= 83.2: # V
> I[1,n]:= 28.7: # A
> phi[n]:= arccos(0.844):
> evalf(180/Pi*phi[n]); # Grad
```

$$32.43504807$$

```
> n[0]:= 1000:   # 1/min
> R[1]:= 0.465:    # Ohm
> R[2]:= 0.00985:  # Ohm
> ü:= U[n]/U[2,n]; # Übersetzungsverhältnis
```

$$\ddot{u} := 6.010$$

Leerlauf:

```
> I[1,0]:= 8.5: # A
> phi[0]:= arccos(0.15):
> evalf(180/Pi*%); # phi[0] in Grad
```

$$81.37307342$$

Kurzschlussversuch:

```
> U[k]:= 170.5: # V
> I[k]:= 28.7:  # A
> phi[k]:= arccos(0.277):
> evalf(180/Pi*%); # phi[k] in Grad
```

$$73.91876337$$

1 Berechnung der Punkte P_n, P_0 und P_k des Kreisdiagramms

Im Folgenden werden indizierte Maple-Variablen, die Zeigergrößen repräsentieren, durch einen

Unterstrich zwischen Variablennamen und Index gekennzeichnet. Da negative Indizes ausgeschlossen sind, kann die gewählte Darstellung nicht zu Missverständnissen führen. Einen Unterstrich hinter der Index-Angabe lässt Maple nicht zu und ein Unterstrich als erstes Zeichen eines Namens soll Variablen der Maple-Bibliothek vorbehalten bleiben.
Einige Teile von Befehlen werden mit Apostroph-Zeichen eingeschlossen, um deren sofortige Auswertung unter Verwendung vorher zugewiesener Variablenwerte zu unterbinden.

Nennpunkt P_n

Zeiger $\underline{I}_{1,n}$:

```
> I_[1,n]:= 'I[1,n]*exp(j*phi[n])';
```

$$I_{\_1,n} := I_{1,n}\,e^{j\,\phi_n}$$

Zeiger auf den Punkt P_n

```
> Pn_:= I_[1,n]:
```

Leerlaufpunkt P_0

Zeiger $\underline{I}_{1,0}$ und Zeiger auf den Punkt P_0

```
> I_[1,0]:= 'I[1,0]*exp(j*phi[0])';
```

$$I_{\_1,0} := I_{1,0}\,e^{j\,\phi_0}$$

```
> P0_:= I_[1,0]:
```

Kurzschlusspunkt P_k

Mit der beim Kurzschlussversuch ermittelten Kurzschlussspannung U_k wird der Effektivwert des Kurzschlussstrom bei Nennspannung bestimmt.

```
> I[1,k]:= 'I[1,n]*U[n]/U[k]';
```

$$I_{1,k} := \frac{I_{1,n}\,U_n}{U_k}$$

Zeiger des Kurzschlussstroms $\underline{I}_{1,k}$ und Zeiger auf den Punkt P_k:

```
> I_[1,k]:= 'I[1,k]*exp(j*phi[k])';
```

$$I_{\_1,k} := I_{1,k}\,e^{j\,\phi_k}$$

```
> Pk_:= I_[1,k]:
```

2 Bestimmung der Kreisparameter

Bestimmungsgleichungen für die Kreisparameter:
Es sind xm der Abszissenwert und ym der Ordinatenwert des Kreismittelpunktes sowie r der Radius des Kreises.

```
> G1:= '(Im(P0_)-xm)^2+(Re(P0_)-ym)^2 = r^2';
```

$$G1 := (\Im(P0\_) - xm)^2 + (\Re(P0\_) - ym)^2 = r^2$$

```
> G2:= '(Im(Pk_)-xm)^2+(Re(Pk_)-ym)^2 = r^2';
```

$$G2 := (\Im(Pk\_) - xm)^2 + (\Re(Pk\_) - ym)^2 = r^2$$

```
> G3:= '(Im(Pn_)-xm)^2+(Re(Pn_)-ym)^2 = r^2';
```

$$G3 := (\Im(Pn\_) - xm)^2 + (\Re(Pn\_) - ym)^2 = r^2$$

Ermittlung der Lösungen des Gleichungssystems:

```
> Loe:= solve({G1,G2,G3}, [xm,ym,r]);
```

$$Loe := [[xm = 47.827, ym = 1.806, r = 39.427], [xm = 47.827, ym = 1.806, r = -39.427]]$$

```
> assign(Loe);
```

3 Darstellung von Heylandkreis und charakteristischen Punkten

Definition von Prozeduren

Für die Darstellung der komplexen Zeiger als Linie wird die Prozedur *linie* definiert. Diese erzeugt eine Plot-Struktur für eine Linie zwischen Anfangs- und Endpunkt eines Zeigers. Dabei werden die Punkte als komplexe Größen vorgegebenen, deren Imaginärteil den Abszissenabschnitt und deren Realteil den Ordinatenwert bezeichnet.

```
> linie:= proc(ap, ep, opt)
    # ap... Anfangspunkt, ep...Endpunkt
    # opt...Optionen (z.B. Farbe, Dicke); als Liste vorzugeben
    plot([[Im(ap),Re(ap)],[Im(ep),Re(ep)]],op(opt));
  end proc:
```

Die Prozedur *punkt* dient der graphischen Darstellung eines Punktes am Endpunkt eines Zeigers.

```
> punkt:= proc(pp)
    # pp... Zeiger auf den Punkt
    pointplot([[Im(pp), Re(pp)]], symbol=solidcircle):
  end proc:
```

Plot-Strukturen für den Heylandkreis und die Punkte P_0, P_n und P_k erzeugen:

```
> kreis:= plottools[circle]([xm,ym], r, color=black):
> P0:= punkt(P0_):  Pk:= punkt(Pk_):  Pn:= punkt(Pn_):
```

Mittelpunkt des Kreises:

```
> MP:= punkt(ym + j*xm):
```

Konstruktion von Drehmoment- und Leistungslinie

Das Lot vom Punkt P_k schneidet die Abszissenparallele durch P_0 im Punkt D. Für den Zeiger vom Koordinatenursprung auf den Punkt D folgt demnach

```
> D_:= 'Re(P0_)+j*Im(Pk_)';
```

$$D\_ := \Re(P0\_) + \mathrm{j}\,\Im(Pk\_)$$

Die Strecke D–Pk

Im Punkt P_k wird keine mechanische Leistung abgegeben. Die Strecke D–P_k ist daher ein Maß für die

Verlustleistung bei stillstehendem Rotor (Schlupf s = 1). Diese setzt sich im Wesentlichen aus den Stromwärmeverlusten in Stator und Rotors zusammen. Durch Aufteilung der Strecke D–P_k gemäß den Anteilen von Stator und Rotor an der Gesamtheit der Stromwärmeverlustleistung im Kurzschlusspunkt ergibt sich der Punkt E. Die Aufteilung erfolgt unter den in [2] beschriebenen Annahmen nach dem Anteil des Widerstands R_1 und des auf die Primärseite bezogenen Rotorwiderstands R'_2 am Gesamtwiderstand. Es ergeben sich so die Teilstrecke D-E, die den Verlustanteil des Stators bezeichnet, und die Teilstrecke E-P_k für die Stromwärmeverlustleistung im Rotor.

Zeiger des auf Primärseite reduzierten Rotorstroms im Kurzschlusspunkt:

```
> I_[2,k]:= 'I_[1,k]-I_[1,0]';
```

$$\underline{I}_{2,\,k} := \underline{I}_{1,\,k} - \underline{I}_{1,\,0}$$

Strecke D-E:

```
> D_E:= 'Re(I_[2,k])*R[1]/(R[1]+R[2]*ü^2)';
```

$$D\_E := \frac{\Re(\underline{I}_{2,\,k})\, R_1}{\ddot{u}^2\, R_2 + R_1}$$

Zeiger vom Koordinatenursprung auf den Punkt E:

```
> E_:= 'D_ + D_E';
```

$$E\_ := D\_ + D\_E$$

Zeiger vom Punkt P_0 auf den Punkt E:

```
> EE_:= 'E_ - P0_';
```

$$EE\_ := E\_ - P0\_$$

Drehmoment- und Leistungslinie

Die Gerade vom Leerlaufpunkt P_0 durch den Punkt E wird als **Drehmomentlinie** bezeichnet. Sie schneidet den Heylandkreis im Unendlichkeitspunkt P_∞, d. h. an dem Punkt, dem der Schlupf mit dem Wert ∞ zugeordnet ist.

```
> ML:= linie(P0_, E_, [legend="Drehmomentlinie ", color=blue]):
> ML2:= linie(P0_, E_, [color=blue]):
```

Die Gerade von P_0 nach P_k ist die **Leistungslinie.**

```
> LL:= linie(P0_, Pk_, [linestyle=dash, legend="Leistungslinie ",
  color=blue]):
> LL2:= linie(P0_, Pk_, [linestyle=dash, color=blue]):
```

Unter Verwendung der oben definierten Zeiger werden nun noch Plotstrukturen für die Darstellung weiterer charakteristischer Punkte und Linien im Kreisdiagramm erzeugt. Plotstruktur für die graphische Darstellung von Punkt D:

```
> DD:= punkt(D_):  # Name D ist für den Differentialoperator
  reserviert
> E:= punkt(E_):
> P0D:= linie(P0_, D_, [color=black]):
> DPk:= linie(D_, Pk_, [color=black]):
```

Darstellung des Kreisdiagramms

Plot-Struktur für Zeiger $I_{1,0}$:

```
> I_10:= linie(0, P0_, [color=blue]):
```

Konstante für Abstand der Beschriftungen im Kreisdiagramm:

```
> d:= 0.07*Im(Pk_): # Textabstand
> text:= textplot([[Im(P0_)-0.7*d, Re(P0_)+d/2,"P0"], [Im(Pn_)-0.7*
  d, Re(Pn_)+d/2,"Pn"], [Im(Pk_)+d, Re(Pk_),"Pk"], [Im(D_)+d/2, Re
  (D_)+1,"D"], [Im(E_)+d/2, Re(E_)+1,"E"]]):
> Titel:= title=typeset(Typ, "\n U = ",U[n]," V,  I = ",I[1,n],"
  A\n"), titlefont=[TIMES,12,BOLD]:
> KD:= display(kreis, P0, Pk, Pn, DD, E, I_10, P0D, DPk, text,
  scaling=constrained, Titel):
> display(KD,LL, ML, labels=[typeset(Im('I[1]/A')),typeset(Re('I[1]
  /A'))]);
```

Asynchronmotor mit Schleifringläufer, 18 kW
U = 500 V, I = 28.700 A



4 Ermittlung des Kipppunktes; Vervollständigung des Kreisdiagramms

Den Kipppunkt, d. h. den Punkt des maximalen Drehmoments, erhält man, wenn man eine Parallele zur

Drehmomentlinie als Tangente an den oberen Halbkreis legt: Berührungspunkt = Kipppunkt.

Steigung der Drehmomentlinie = Re(EE_)/Im(EE_); EE_ ... Zeiger von P_0 auf den Punkt E

Gleichung einer Tangente am Punkt (x_k, y_k) des Kreises:

```
> G4:= ''(x-xm)*(xk-xm) + (y-ym)*(yk-ym) = r^2'';
```

$$G4 := '(x-xm)\ (xk-xm) + (y-ym)\ (yk-ym) = r^2'$$

Auflösung der Gleichung $G4$ nach y; Erzeugung der Form $y=A \cdot x + B$:

```
> G5:= isolate(G4,y);
```

$$G5 := y = \frac{r^2 - (x-xm)\ (xk-xm)}{yk-ym} + ym$$

Nachdem durch die Apostrophierung in G4 das Einsetzen der Werte von xm und ym zweimal unterbunden wurde, erscheinen diese Werte jetzt in der Rechnung.

```
> G6:= collect(G5, x);
```

$$G6 := y = \frac{(-1.000\ xk + 47.827)\ x}{yk - 1.806} + \frac{-732.953 + 47.827\ xk}{yk - 1.806} + 1.806$$

Terme des Ausdrucks zur Berechnung von y ermitteln:

```
> AA:= op(rhs(G6));
```

$$AA := \frac{(-1.000\ xk + 47.827)\ x}{yk - 1.806},\ \frac{-732.953 + 47.827\ xk}{yk - 1.806},\ 1.806$$

Der 1. Term von AA enthäht den Koeffizienten von x. Herauslösen des Koeffizienten aus AA[1], Gleichsetzung mit der Steigung der Drehmomentlinie und Auflösung nach yk:

```
> A := AA[1]/x;
```

$$A := \frac{-1.000\ xk + 47.827}{yk - 1.806}$$

```
> G7:= isolate(A = Re(EE_)/Im(EE_), yk);
```

$$G7 := yk = -5.804\ xk + 279.383$$

Die Kreisgleichung lautet

```
> G8:= (xk-xm)^2 + (yk-ym)^2 = r^2;
```

$$G8 := (xk - 47.827)^2 + (yk - 1.806)^2 = 1554.466$$

Das Gleichungssystem {G6, G7} hat zwei Lösungen:

```
> Loe2:= solve({G7,G8}, [xk,yk]);
```

$$Loe2 := [[xk = 41.132, yk = 40.660], [xk = 54.522, yk = -37.048]]$$

Nur die erste Lösung entspricht dem gesuchten Punkt P_{kipp}:

Zeiger auf P_{kipp}:

```
> I_[1,kipp]:= subs(Loe2[1], j*xk + yk);
```

$$I\_{1,\ kipp} := 40.660 + 41.132\ \mathrm{j}$$

```
> Pkipp_:= I_[1,kipp]:
```

Plot-Struktur für Kipppunkt P_{kipp} erzeugen:

```
> Pkipp:= punkt(Pkipp_):
```

Vervollständigung des Kreisdiagramms:

```
> text2:= textplot([Im(Pkipp_),Re(Pkipp_)+d,"Pkipp"]):
> KD2:= display(KD,Pkipp,text2):
> display(KD2,LL, ML, labels=[typeset(Im('I[1]/A')),typeset(Re('I[1]/A'))]);
```



5 Rechnerische Auswertung des Kreisdiagramms

Bestimmung der Maßstäbe für die Leistung und das Drehmoment

```
> LMS:= sqrt(3)*U[n]/1000; # Leistungsmaßstab in kW/A
```

$$LMS := \frac{1}{2}\sqrt{3}$$

```
> DMS:= sqrt(3)*U[n]/(2*Pi*n[0]/60); # Drehmomentmaßstab in Nm/A
```

$$DMS := \frac{15\sqrt{3}}{\pi}$$

Prozedur zur Ermittlung des Drehmoments am Punkt P:

```
> moment:= proc(P_)
```

```
    local S;  global P0_, E_, D_, DMS;
    # Ermittlung der Strecke von P bis zur Drehmomentlinie
    S:= Re(P_)-Re(P0_)-(Im(P_)-Im(P0_))*(Re(E_)-Re(D_))/(Im(D_)-Im
  (P0_));
    evalf(S*DMS);
  end proc:
```

Prozedur zur Ermittlung der mechanischen Leistung am Punkt P:

```
> leistung:= proc(P_)
    local S;  global P0_, Pk_, D_, LMS;
    # Ermittlung der Strecke von P bis zur Leistungslinie
    S:= Re(P_)-Re(P0_)-(Im(P_)-Im(P0_))*(Re(Pk_)-Re(D_))/(Im(D_)-Im
  (P0_));
    evalf(S*LMS);
  end proc:
```

Auswertung für den Nennpunkt P_n:

Drehmoment in Nm:

```
> M[n]:= moment(Pn_);
```

$$M_n := 179.818$$

Mechanische Leistung in kW:

```
> Pmech[n]:= leistung(Pn_);
```

$$Pmech_n := 18.033$$

Auswertung für den Kipppunkt P_{kipp}:

Drehmoment in Nm:

```
> M[kipp]:= moment(Pkipp_);
```

$$M_{kipp} := 279.078$$

Mechanische Leistung in kW:

```
> Pmech[kipp]:= leistung(Pkipp_);
```

$$Pmech_{kipp} := 25.489$$

6 Drucken des vervollständigten Kreisdiagramms

Das Diagramm wird im Postskriptformat in der Datei *print.eps* gespeichert um es bei Bedarf ausgedrucken zu können. Mit dem Befehl **plotsetup** wird diese Form der Ausgabe vorbereitet. In Anpassung an die übliche Darstellung des Kreisdiagramms werden die Achsenbezeichnungen und auch die Achsenmarkierungen bei der Ausgabe unterdrückt.

Das Postskriptformat wird wegen der Qualität des Druckbildes bevorzugt. Für das Drucken des Postskriptfiles wurde beim Programmtest des lizenzfreie Programm GSview verwendet.

Wenn während der Ausführung des Programms die Frage "Kreisdiagramm drucken?" mit "ja" beantwortet wird, kommt automatisch das in der Datei print.eps abgelegte Diagramm über GSview (oder ein anderes geeignetes Programm) zur Anzeige und kann ausgedruckt werden. Anschließend wird die Eingabe der Länge der Strecke vom Punkt P_0 zum Punkt P_k angefordert und daraus Strom-, Drehmoment- und Leistungsmaßstab für das gedruckte Diagramm berechnet. Diese Daten werden auf dem Bildschirm angezeigt und außerdem gemeinsam mit dem Kreisdiagramm in die korrigierte Datei print.eps eingetragen, die wiederum angezeigt wird und ausgedruckt werden kann.

```
> plotsetup(ps, plotoutput="print", plotoptions="noborder,
  resolution=2000");
> KD3:= display(KD2, LL2, ML2, font=[TIMES,8], tickmarks=[0,0],
  caption=typeset("\nStrommaßstab: ", IMSd, "
  A/mm\rLeistungsmaßstab: ", LMSd, " kW/mm \nDrehmomentmaßstab: ",
  DMSd, " Nm/mm"), titlefont=[TIMES,12], captionfont=[TIMES,8]):
  KD3;

> druck:= readstat("Kreisdiagramm drucken? ja oder nein"):
> if druck='ja' then
    system[launch]("cmd /c","print.eps");
    ZL:= readstat("Zeigerlänge von Punkt P0 bis Pk in mm:"):
    printf("Zeigerlänge von Punkt P0 bis Pk: %4.1f mm \n", ZL);
    # Ermittlung des Strommaßstabs:
    IMSd:= abs(Pk_-P0_)/ZL; # A/mm
    # Berechnung und Anzeige der Maßstäbe:
    LMSd:= evalf(IMSd*U[n]*sqrt(3)/1000);   # kW/mm
    DMSd:= evalf(LMSd*1000/(2*Pi*n[0]/60)); # Nm/mm
    printf("Strommaßstab:     %6.2f A/mm \n", IMSd);
    printf("Leistungsmaßstab: %6.2f kW/mm \n", LMSd);
    printf("Drehmomentmaßstab:%6.2f Nm/mm \n", DMSd);
    # Ausgabe des ergänzten Kreisdiagramms
    print(KD3); Threads[Sleep](1):
    system[launch]("cmd /c","print.eps");
  end if:
Zeigerlänge von Punkt P0 bis Pk: 126.5 mm
Strommaßstab:       0.60 A/mm
Leistungsmaßstab:   0.52 kW/mm
Drehmomentmaßstab:  4.95 Nm/mm
```